
NAME
perldelta - what is new for perl 5.10.0

DESCRIPTION
This document describes the differences between the 5.8.8 release and
 the 5.10.0 release.

Many of the bug fixes in 5.10.0 were already seen in the 5.8.X maintenance
 releases; they are not 
duplicated here and are documented in the set of
 man pages named perl58[1-8]?delta.

Core Enhancements
The feature pragma

The feature pragma is used to enable new syntax that would break Perl's
 backwards-compatibility 
with older releases of the language. It's a lexical
 pragma, like strict or warnings.

Currently the following new features are available: switch (adds a
 switch statement), say (adds a 
say built-in function), and state
 (adds a state keyword for declaring "static" variables). Those

features are described in their own sections of this document.

The feature pragma is also implicitly loaded when you require a minimal
 perl version (with the use 
VERSION construct) greater than, or equal
 to, 5.9.5. See feature for details.

New -E command-line switch
-E is equivalent to -e, but it implicitly enables all
 optional features (like use feature ":5.10").

Defined-or operator
A new operator // (defined-or) has been implemented.
 The following expression:

    $a // $b

is merely equivalent to

   defined $a ? $a : $b

and the statement

   $c //= $d;

can now be used instead of

   $c = $d unless defined $c;

The // operator has the same precedence and associativity as ||.
 Special care has been taken to 
ensure that this operator Do What You Mean
 while not breaking old code, but some edge cases 
involving the empty
 regular expression may now parse differently. See perlop for
 details.

Switch and Smart Match operator
Perl 5 now has a switch statement. It's available when use feature
 'switch' is in effect. This 
feature introduces three new keywords, given, when, and default:

    given ($foo) {
	 when (/^abc/) { $abc = 1; }
	 when (/^def/) { $def = 1; }
	 when (/^xyz/) { $xyz = 1; }
	 default { $nothing = 1; }
    }

A more complete description of how Perl matches the switch variable
 against the when conditions is 
given in "Switch statements" in perlsyn.

This kind of match is called smart match, and it's also possible to use
 it outside of switch statements, 
via the new ~~ operator. See "Smart matching in detail" in perlsyn.

This feature was contributed by Robin Houston.

Regular expressions
Recursive Patterns

It is now possible to write recursive patterns without using the (??{})
 construct. This new 
way is more efficient, and in many cases easier to
 read.

Each capturing parenthesis can now be treated as an independent pattern
 that can be entered
by using the (?PARNO) syntax (PARNO standing for
 "parenthesis number"). For example, the 
following pattern will match
 nested balanced angle brackets:

    /
     ^                      # start of line
     (                      # start capture buffer 1
	 <                   #   match an opening angle bracket
	 (?:                 #   match one of:
	    (?>             #     don't backtrack over the inside of this 
group
		 [^<>]+      #       one or more non angle brackets
	    )               #     end non backtracking group
	 |                   #     ... or ...
	    (?1)            #     recurse to bracket 1 and try it again
	 )*                  #   0 or more times.
	 >                   #   match a closing angle bracket
     )                      # end capture buffer one
     $                      # end of line
    /x

PCRE users should note that Perl's recursive regex feature allows
 backtracking into a 
recursed pattern, whereas in PCRE the recursion is
 atomic or "possessive" in nature. As in the
example above, you can
 add (?>) to control this selectively. (Yves Orton)

Named Capture Buffers

It is now possible to name capturing parenthesis in a pattern and refer to
 the captured 
contents by name. The naming syntax is (?<NAME>....).
 It's possible to backreference to a 
named buffer with the \k<NAME>
 syntax. In code, the new magical hashes %+ and %- can be 
used to
 access the contents of the capture buffers.

Thus, to replace all doubled chars with a single copy, one could write

    s/(?<letter>.)\k<letter>/$+{letter}/g

Only buffers with defined contents will be "visible" in the %+ hash, so
 it's possible to do 
something like

    foreach my $name (keys %+) {
        print "content of buffer '$name' is $+{$name}\n";
    }

The %- hash is a bit more complete, since it will contain array refs
 holding values from all 
capture buffers similarly named, if there should
 be many of them.

%+ and %- are implemented as tied hashes through the new module 
Tie::Hash::NamedCapture.




